

五極管の等価回路

Ayumi's Lab.

2004年5月27日

Revised 2005年2月16日

目次

1 等価回路	1
1.1 数値例	4
1.2 シミュレーション	5
1.3 ロードライン	7
2 より精緻な等価回路	7
2.1 数値例	10
2.2 等価回路	10
3 スクリーニンググリッドパスコンによる周波数特性	10
4 QUAD II の位相反転回路	12
4.1 スクリーニンググリッド結合のみ	12
4.1.1 数値例	13
4.1.2 シミュレーション	13

1 等価回路

五極管は、通常、スクリーングリッド (G2) とカソードを交流的に短絡して使用します。

電圧増幅回路では、B 電源より抵抗 R_{g2} を介して G2 に電圧を加え、G2 と K の間にバイパスコンデンサ C_{g2} を入れます。こうすることにより、最大の増幅度が得られます (図 1)。このときの中域における等価回路は、電流源による三極管の等価回路と等しくなります (図 2)。等価回路より、

$$e_p = -g_m e_g (r_p // R_L) \quad (1)$$

$$A = -g_m (r_p // R_L) \quad (2)$$

となります。

G2 と K の間にパスコンがない場合、G2 に流れる電流 i_{g2} によって R_{g2} の両端に信号電圧が発生し、G2 の電圧が変動することから、一種の負帰還がかかり、増幅度が下がります。これを解析するためには、図 3 のような等価回路を使用するとよいでしょう。ここで、




図 1: 五極管によるカソード接地増幅回路




図 2: 五極管によるカソード接地増幅回路の等価回路 (1)




図 3: 五極管によるカソード接地増幅回路の等価回路 (2)

$$g_{mg2} = \text{第2グリッド相互コンダクタンス } \frac{\partial I_{g2}}{\partial E_g}$$

$$\approx \frac{I_{g2}}{I_p} g_m \quad (3)$$

$$r_{g2} = \text{第2グリッド内部抵抗 } \frac{\partial E_{g2}}{\partial I_{g2}}$$

$$\approx \frac{I_p + I_{g2}}{I_{g2}} r_{p(T)} \quad (4)$$

$$\frac{g_m}{g_{mg2}} \approx \frac{I_p}{I_{g2}} \quad (5)$$

で、(T) が付いた三定数は、プレート電圧が E_{g2} のときの三極管接続時のものです。

この等価回路の前提となっているのは、

五極管では、プレート電圧が変化しても、プレート電流 I_p と第2グリッド電流 I_{g2} の比は一定である

というのですが、実際には、プレート電圧が下がれば第2グリッド電流の比率が増えるので、2番目の電流源（電流制御電流源）のゲインは g_m/g_{mg2} 一定ではありません。

この等価回路より、

$$e_{g2} = -g_{mg2} e_g (r_{g2} // R_{g2})$$

$$= -\frac{I_{g2}}{I_p} g_m e_g \left\{ \frac{I_p + I_{g2}}{I_{g2}} r_{p(T)} // R_{g2} \right\} \quad (6)$$

$$i_{g2} = \frac{-e_{g2}}{R_{g2}}$$

$$= g_{mg2} e_g \frac{r_{g2} // R_{g2}}{R_{g2}} = g_{mg2} e_g \frac{r_{g2}}{r_{g2} + R_{g2}} \quad (7)$$

$$e_p = -\frac{g_m}{g_{mg2}} i_{g2} (r_p // R_L)$$

$$= -\frac{g_m}{g_{mg2}} g_{mg2} e_g \frac{r_{g2}}{r_{g2} + R_{g2}} (r_p // R_L)$$

$$= -g_m e_g \frac{r_{g2}}{r_{g2} + R_{g2}} (r_p // R_L) \quad (8)$$

$$A_f = -g_m \frac{r_{g2}}{r_{g2} + R_{g2}} (r_p // R_L) \quad (9)$$

これらの式で、 $R_{g2} \rightarrow 0$ とすれば、G2 をバイパスした通常の五極管の式と一致します。

パスコンがないことによるゲインの低下を負帰還と考えると、

$$A_f = \frac{A}{1 + A\beta} = \frac{g_m (r_p // R_L)}{1 + \frac{R_{g2}}{r_{g2}}}$$

より、帰還量 $1 + A\beta$ は、

$$1 + A\beta = 1 + \frac{R_{g2}}{r_{g2}} \quad (10)$$

帰還率 β は、

$$\beta = \frac{R_{g2}}{r_{g2} g_m (r_p // R_L)} \quad (11)$$

となります。

また、図4のような等価回路を考えることもできます。




図 4: 五極管によるカソード接地増幅回路の等価回路 (3)

この等価回路より，次の関係が成り立つことがわかります．

$$e_{g2} = -g_{mg2}e_g(r_{g2}/R_{g2}) \quad (12)$$

$$e_p = -g_m\left(e_g + \frac{e_{g2}}{\mu_{g1-g2}}\right)(r_p//R_L) \quad (13)$$

これより，

$$\begin{aligned} e_p &= -g_m\left\{e_g - \frac{g_{mg2}e_g(r_{g2}/R_{g2})}{\mu_{g1-g2}}\right\}(r_p//R_L) \\ &= -g_me_g\left\{1 - \frac{g_{mg2}}{\mu_{g1-g2}} \cdot \frac{r_{g2}R_{g2}}{r_{g2} + R_{g2}}\right\}(r_p//R_L) \\ &= -g_me_g\left\{1 - \frac{1}{r_{g2}} \cdot \frac{r_{g2}R_{g2}}{r_{g2} + R_{g2}}\right\}(r_p//R_L) \\ &= -g_me_g\left\{1 - \frac{R_{g2}}{r_{g2} + R_{g2}}\right\}(r_p//R_L) \\ &= -g_me_g \frac{r_{g2}}{r_{g2} + R_{g2}}(r_p//R_L) \end{aligned} \quad (14)$$

となって，式 (8) と一致します．

1.1 数値例

QUAD II の初段の片側を使って計算してみます．定数は以下の通りです．

$$E_{bb} = 328 \text{ V}$$

$$E_p = 132.6 \text{ V}$$

$$E_{g2} = 97.13 \text{ V}$$

$$E_g = -2.008 \text{ V}$$

$$R_L = 180//680 = 142.3 \text{ k}\Omega$$

$$R_{g2} = 1 \text{ M}\Omega$$

$$I_p = 1.086 \text{ mA}$$

$$I_{g2} = 0.231 \text{ mA}$$

$$g_m = 1.286 \text{ mS}$$

$$g_{mg2} = 0.2735 \text{ mS}$$

$$r_p = 3009 \text{ k}\Omega$$

$$r_{g2} = 133.3 \text{ k}\Omega$$

$$\begin{aligned}\mu_{g1-g2} &= 36.57 \\ r_{p(T)} &= 23.44 \text{ k}\Omega \\ \mu_{(T)} &= 36.45\end{aligned}$$

これより，パスコンありのゲインは，

$$A = -g_m(r_p//R_L) = -1.286(3009//142.3) = -174.8 \quad (15)$$

パスコンなしのゲインは，

$$A_f = -g_m \frac{r_{g2}}{r_{g2} + R_{g2}} (r_p//R_L) = -1.286 \frac{133.3}{133.3 + 1000} (3009//142.3) = -20.55 \quad (16)$$

となります．シミュレーションによるゲインは，それぞれ 174.8 と 18.8 でした．後者の誤差は約 9.3% です．

1.2 シミュレーション

6267.cir

```
1 6267 voltage amp
2 .INCLUDE 6267.lib
3
4 XV1 1 2 3 4 6267
5 VIP 11 1 0
6 VIG2 12 2 0
7 RP 5 11 180k
8 RG2 5 12 1Meg
9 CG2 12 0 1u
10 RK 4 0 1525
11 CK 4 0 470u
12 RG 3 0 470k
13 RGx 6 0 680k
14 CC 1 6 1u
15 VB 5 0 330
16 VI 3 0 DC 0 AC 1
17
18 .control
19 ac dec 10 1k 1k
20 print abs(v(6)) abs(v(2))
21 .endc
22 .END
```

6267.lib

```
1 *
2 * GENERIC: 6267
3 *
4 .SUBCKT 6267 A G2 G1 K
5 BGG GG 0 V=V(G1,K)+0.60876
6 BEP EP 0 V=URAMP(V(A,K))+1e-10
7 BEG2 EG2 0 V=URAMP(V(G2,K))+1e-10
8 BSTM STM 0 V=URAMP(V(GG)+V(EG2)/31.574)+1e-10
9 BM1 M1 0 V=(0.00776635839614873*(URAMP(V(EG2)-1e-10)+1e-10))^-0.487320892704545
10 BM2 M2 0 V=(0.754785*V(STM))^1.98732089270454
11 BM M 0 V=V(M1)*V(M2)
12 BSTP STP 0 V=URAMP(V(GG)+V(EG2)/41.8317799108355)+1e-10
13 BP P 0 V=1.72960609602883*V(STP)^1.5
14 BIK IK 0 V=U(V(GG))*V(P)+(1-U(V(GG)))*V(M)
15 BLIM LI 0 V=0.88*V(EP)^1.5
16 BEG EG 0 V=URAMP(V(G1,K))+1e-10
```

```

17 BIG IG 0 V=0.88*(V(EG)/(V(EP)+V(EG))*1.2+0.4)*V(EG)^1.5
18 BIK2 IK2 0 V=V(IK,IG)*(1-0.4*(EXP(-V(EP)/V(EG2)*15)-EXP(-15)))
19 BIG2T IG2T 0 V=V(IK2)*(0.83967*(1-V(EP)/(V(EP)+10))^1.5+0.16033)
20 BIK3 IK3 0 V=V(IK2)*(V(EP)+11110)/(V(EG2)+11110)
21 BKLIM KLI 0 V=0.88*(V(EP)+URAMP(V(EG2,EP)))^1.5
22 BIK4 IK4 0 V=V(IK3)-URAMP(V(IK3,KLI))
23 BIP IP 0 V=URAMP(V(IK4,IG2T))-URAMP(V(IK4,IG2T)-V(LI)))
24 BIAK A K I=0.00071561*V(IP)
25 BIG2 G2 K I=0.00071561*V(IK4,IP)
26 BIGK G1 K I=0.00071561*V(IG)
27 * CAPS
28 CAK A K 5.1e-12
29 CGK G1 K 3.8e-12
30 CGA G1 A 5e-14
31 .ENDS

```

結果 (パソコンあり)

```

1
2 Circuit: 6267 voltage amp
3
4 abs(v(6)) = 1.747665e+02
5 abs(v(2)) = 4.951569e-02

```

結果 (パソコンなし)

```

1
2 Circuit: 6267 voltage amp
3
4 abs(v(6)) = 1.879167e+01
5 abs(v(2)) = 3.263731e+01

```

パソコンなしの場合の、各点の電圧・電流は、以下の通りです。

	$e_g = 0$	$e_g = 0.1 \text{ V}$	Δ	小信号解析
E_g	-2.00755 V	-1.90755 V	0.1 V	—
E_p	132.5911 V	130.7134 V	-1.8777 V	-18.76174
I_p	1.085563 mA	1.098756 mA	0.013193 mA	0.1318227
E_{g2}	97.13170 V	93.85699 V	-3.27471 V	-32.85897
I_{g2}	0.2308608 mA	0.2341355 mA	0.0032747 mA	0.03285897

1.3 ロードライン




図 5: 6267 のロードライン

2 より精緻な等価回路

では、特性曲線を詳しく調べ、どこで誤差が出ているか検討しましょう。図 6 は、 $E_p = 133.826 \text{ V}$ の、 $E_{g2}-I_{g2}$ 特性を拡大したものです。O が無信号時の動作点です。

$e_i = 0.1 \text{ V}$ を加えると、動作点は A' に移動するように思われます。しかし、プレートに信号が発生し、プレート電圧が下がります。プレートに発生した信号の大きさを（この段階ではわからないのですが） e_p とします。プレート電圧が下がると第 2 グリッドに流れる電流がわずかに増え、特性曲線は緑色のようになります。したがって、動作点は A' ではなく、A に移動します。

それでは、この点 A を求めます。OD が求めるべき第 2 グリッドの信号出力です。

$$OD = e_{g2} \tag{17}$$

BO は、増幅率の定義より、

$$BO = \mu_{g1-g2} e_g \tag{18}$$

これらより、

$$\begin{aligned} BD + DO &= BO \\ BD - e_{g2} &= \mu_{g1-g2} e_g \\ BD &= \mu_{g1-g2} e_g + e_{g2} \end{aligned} \tag{19}$$

内部抵抗の定義より、

$$r_{g2} = \frac{BD}{DC}$$




図 6: 6267 の $E_{g2}-I_{g2}$ 特性

$$\begin{aligned}
 DC &= \frac{BD}{r_{g2}} = \frac{\mu_{g1-g2}e_g + e_{g2}}{r_{g2}} \\
 &= g_{mg2}e_g + \frac{e_{g2}}{r_{g2}} \quad (20)
 \end{aligned}$$

CA は、第 1 グリッド電圧および第 2 グリッド電圧が一定の場合に、プレート電圧を変化させたときの I_{g2} の変化率から求めます。

$$CA = \frac{\partial I_{g2}}{\partial E_p} e_p \quad (21)$$

ここでは、プレート電圧が下がったときにプレート電流が減る分の f 倍が第 2 グリッド電流の増加分になると仮定します。6267 のこの動作点の場合、 $f = 0.6477$ です。このとき、

$$CA = -f \frac{e_p}{r_p} \quad (22)$$

となります。

三角形 DAO より、

$$\begin{aligned}
 R_{g2} &= \frac{DO}{DA} = \frac{-e_{g2}}{DC + CA} \\
 e_{g2} &= -R_{g2}(DC + CA) \\
 &= -R_{g2}\left(g_{mg2}e_g + \frac{e_{g2}}{r_{g2}} - f \frac{e_p}{r_p}\right) \\
 \frac{e_{g2}}{R_{g2}} &= g_{mg2}e_g + \frac{e_{g2}}{r_{g2}} - f \frac{e_p}{r_p} \\
 \left(\frac{1}{r_{g2}} + \frac{1}{R_{g2}}\right) &= -g_{mg2}e_g + f \frac{e_p}{r_p} \\
 e_{g2} &= \left(-g_{mg2}e_g + f \frac{e_p}{r_p}\right)(r_{g2} // R_{g2}) \quad (23)
 \end{aligned}$$

今度は、プレート側をみます。図は 7 です。




図 7: 6267 の E_p - I_p 特性

$$OF = g_m e_g \quad (24)$$

$$FG = g_m \frac{e_{g2}}{\mu_{g1-g2}} \quad (25)$$

$$OG = OF + FG = g_m \left(e_g + \frac{e_{g2}}{\mu_{g1-g2}} \right) \quad (26)$$

第 2 グリッドが 1V 上昇すると，第 1 グリッドが $1/\mu_{g1-g2}$ V 上昇したのと等価なことに注意してください．これより，

$$\begin{aligned} e_p &= OI = -OG(r_p//R_L) \\ &= -g_m \left(e_g + \frac{e_{g2}}{\mu_{g1-g2}} \right) (r_p//R_L) \end{aligned} \quad (27)$$

この式に式 (23) を代入して，

$$\begin{aligned} e_p &= -g_m \left\{ e_g + \left(-g_{mg2} e_g + f \frac{e_p}{r_p} \right) \frac{r_{g2} // R_{g2}}{\mu_{g1-g2}} \right\} (r_p // R_L) \\ \left(1 + \frac{g_m}{\mu_{g1-g2}} (r_{g2} // R_{g2}) \frac{f}{r_p} (r_p // R_L) \right) e_p &= -g_m e_g \left(1 - \frac{g_{mg2}}{\mu_{g1-g2}} (r_{g2} // R_{g2}) \right) (r_p // R_L) \\ e_p &= -g_m e_g (r_p // R_L) \frac{1 - \frac{g_{mg2}}{\mu_{g1-g2}} (r_{g2} // R_{g2})}{1 + \frac{g_m}{\mu_{g1-g2}} (r_{g2} // R_{g2}) f \frac{R_L}{r_p + R_L}} \\ &= -g_m e_g (r_p // R_L) \frac{1 - \frac{g_{mg2}}{\mu_{g1-g2}} (r_{g2} // R_{g2})}{1 + f \frac{R_L}{r_p + R_L} \cdot \frac{I_{p0}}{I_{g20}} \cdot \frac{g_{mg2}}{\mu_{g1-g2}} (r_{g2} // R_{g2})} \end{aligned} \quad (28)$$

$A_g = g_{mg2}(r_{g2} // R_{g2}) / \mu_{g1-g2}$ とおくと，

$$e_p = -g_m e_g (r_p // R_L) \frac{1 - A_g}{1 + f \frac{R_L}{r_p + R_L} \cdot \frac{I_{p0}}{I_{g20}} A_g} \quad (29)$$

$$A_f = -g_m (r_p // R_L) \frac{1 - A_g}{1 + f \frac{R_L}{r_p + R_L} \cdot \frac{I_{p0}}{I_{g20}} A_g} \quad (30)$$

となります。

2.1 数値例

$$A_g = \frac{0.231}{1.086} \times 1.286 \times (133.3//1000) = 0.879467$$

$$A_f = -1.286 \times (3009//142.3) \times \frac{1 - 0.879467}{1 + 0.6477 \times \frac{142.3}{3009+142.3} \times \frac{1.086}{0.231} \times 0.879467} = -18.79$$

となり、シミュレーションの結果とほとんど一致します。

2.2 等価回路

これまでの考察から得られる等価回路は、図 8 のようになります。一番左の電流源は、G1 によ




図 8: 五極管によるカソード接地増幅回路の等価回路 (4)

る G2 電流の変化を表し、2 番目の電流源は、プレート電圧の変化による G2 電流の変化を表し、3 番目の電流源は、G1 および G2 によるプレート電流の変化を表しています。

3 スクリーニンググリッドパスコンによる周波数特性

通常、スクリーングリッドにはパスコンを付けますが、低域ではインピーダンスが上昇し、パイパスの役目を果たさなくなります。中域では式 (2) のゲインであったものが、最終的には、式 (30) のゲインまで下がります。

このカットオフ周波数を求めてみましょう。簡単にするため、図 4 の等価回路をベースにします。これに、G2 のパスコンを加えた等価回路は、図 9 のようになります。




図 9: G2 パスコンを考慮した五極管によるカソード接地増幅回路の等価回路

この等価回路より、次の関係が成り立ちます。

$$e_{g2} = -g_{mg2} e_g (r_{g2} // R_{g2} // Z_{C_{g2}}) \quad (31)$$

$$e_p = -g_m \left(e_g + \frac{e_{g2}}{\mu_{g1-g2}} \right) (r_p // R_L) \quad (32)$$

これより,

$$\begin{aligned}
 e_p &= -g_m \left\{ e_g - \frac{g_{m2} e_g (r_{g2} // R_{g2} // Z_{C_{g2}})}{\mu_{g1-g2}} \right\} (r_p // R_L) \\
 &= -g_m e_g \left(1 - \frac{g_{m2}}{\mu_{g1-g2}} \cdot \frac{1}{\frac{1}{r_{g2}} + \frac{1}{R_{g2}} + \frac{1}{Z_{C_{g2}}}} \right) (r_p // R_L) \\
 &= -g_m e_g \left(1 - \frac{1}{r_{g2}} \cdot \frac{1}{\frac{1}{r_{g2}} + \frac{1}{R_{g2}} + \frac{1}{Z_{C_{g2}}}} \right) (r_p // R_L) \\
 &= -g_m e_g \left(1 - \frac{1}{1 + \frac{r_{g2}}{R_{g2}} + \frac{r_{g2}}{Z_{C_{g2}}}} \right) (r_p // R_L) \\
 &= -g_m e_g \left(1 - \frac{R_{g2} Z_{C_{g2}}}{R_{g2} Z_{C_{g2}} + r_{g2} Z_{C_{g2}} + r_{g2} R_{g2}} \right) (r_p // R_L) \\
 &= -g_m e_g \frac{r_{g2} Z_{C_{g2}} + r_{g2} R_{g2}}{R_{g2} Z_{C_{g2}} + r_{g2} Z_{C_{g2}} + r_{g2} R_{g2}} (r_p // R_L) \\
 &= -g_m e_g \frac{r_{g2} Z_{C_{g2}} + r_{g2} R_{g2}}{R_{g2} Z_{C_{g2}} + r_{g2} Z_{C_{g2}} + r_{g2} R_{g2}} (r_p // R_L) \\
 &= -g_m e_g \frac{1 + \frac{Z_{C_{g2}}}{R_{g2}}}{1 + Z_{C_{g2}} \left(\frac{1}{r_{g2}} + \frac{1}{R_{g2}} \right)} (r_p // R_L) \\
 &= -g_m e_g \frac{1 + \frac{1}{j\omega C_{g2} R_{g2}}}{1 + \frac{1}{j\omega C_{g2} (r_{g2} // R_{g2})}} (r_p // R_L)
 \end{aligned}$$

中域のゲインを $A_M = -g_m (r_p // R_L)$ とすると, 任意の周波数のゲイン A は,

$$A = \frac{1 + \frac{1}{j\omega C_{g2} R_{g2}}}{1 + \frac{1}{j\omega C_{g2} (r_{g2} // R_{g2})}} A_M \quad (33)$$

と表せます. この式の分子はカットオフ周波数 $f_1 = 1/(2\pi C_{g2} R_{g2})$ の低域上昇を表し, 分母はカットオフ周波数 $f_2 = 1/\{2\pi C_{g2} (r_{g2} // R_{g2})\}$ の低域下降を表していますから, 周波数特性は, 図 10 のようになります.




図 10: G2 パスコンによる周波数特性

4 QUAD IIの位相反転回路

4.1 スクリーニンググリッド結合のみ

スクリーングリッド結合のみの等価回路は，図 11 のようになります．




図 11: スクリーングリッド結合位相反転回路の等価回路

等価回路より，以下の関係が成り立ちます．

$$e_{g2} = -(g_{m2}e_g - fi_{rp1} - fi_{rp2}) \frac{r_{g2} // R_{g2}}{2} \quad (34)$$

$$e_{p1} = -\left(g_m e_g + g_m \frac{e_{g2}}{\mu_{g1-g2}}\right) (r_p // R_L) \quad (35)$$

$$e_{p2} = -g_m \frac{e_{g2}}{\mu_{g1-g2}} (r_p // R_L) \quad (36)$$

$$i_{rp1} = \frac{e_{p1}}{r_p} \quad (37)$$

$$i_{rp2} = \frac{e_{p2}}{r_p} \quad (38)$$

式 (37), (38) に式 (35), (36) を代入して，

$$\begin{aligned} i_{rp1} &= -\left(g_m e_g + g_m \frac{e_{g2}}{\mu_{g1-g2}}\right) \frac{R_L}{r_p + R_L} \\ &= -\frac{I_p}{I_{g2}} g_{m2} \left(e_g + \frac{e_{g2}}{\mu_{g1-g2}}\right) \frac{R_L}{r_p + R_L} \end{aligned} \quad (39)$$

$$\begin{aligned} i_{rp2} &= -g_m \frac{e_{g2}}{\mu_{g1-g2}} \cdot \frac{R_L}{r_p + R_L} \\ &= -\frac{I_p}{I_{g2}} g_{m2} \frac{e_{g2}}{\mu_{g1-g2}} \cdot \frac{R_L}{r_p + R_L} \end{aligned} \quad (40)$$

式 (34) に式 (39), (40) を代入して，

$$\begin{aligned} e_{g2} &= -\left\{g_{m2}e_g + f \frac{I_p}{I_{g2}} g_{m2} \left(e_g + \frac{2e_{g2}}{\mu_{g1-g2}}\right) \frac{R_L}{r_p + R_L}\right\} \frac{r_{g2} // R_{g2}}{2} \\ &= -\left\{g_{m2}e_g \left(1 + f \frac{I_p}{I_{g2}} \cdot \frac{R_L}{r_p + R_L}\right) + 2e_{g2} f \frac{I_p}{I_{g2}} \cdot \frac{g_{m2}}{\mu_{g1-g2}} \cdot \frac{R_L}{r_p + R_L}\right\} \frac{r_{g2} // R_{g2}}{2} \\ e_{g2} + f \frac{I_p}{I_{g2}} \cdot \frac{g_{m2}}{\mu_{g1-g2}} (r_{g2} // R_{g2}) \frac{R_L}{r_p + R_L} e_{g2} \\ &= -g_{m2}e_g \left(1 + f \frac{I_p}{I_{g2}} \cdot \frac{R_L}{r_p + R_L}\right) \frac{r_{g2} // R_{g2}}{2} \end{aligned}$$

$$\begin{aligned}
e_{g2} &= -g_{mg2}e_g \frac{1 + f \frac{I_p}{I_{g2}} \cdot \frac{R_L}{r_p + R_L}}{1 + f \frac{I_p}{I_{g2}} \cdot \frac{R_L}{r_p + R_L} \cdot \frac{g_{mg2}}{\mu_{g1-g2}} (r_{g2} // R_{g2})} \cdot \frac{r_{g2} // R_{g2}}{2} \\
&= -\frac{g_{mg2}(r_{g2} // R_{g2})e_g}{2} \cdot \frac{1 + f \frac{I_p}{I_{g2}} \cdot \frac{R_L}{r_p + R_L}}{1 + f \frac{I_p}{I_{g2}} \cdot \frac{R_L}{r_p + R_L} \cdot \frac{g_{mg2}}{\mu_{g1-g2}} (r_{g2} // R_{g2})}
\end{aligned} \tag{41}$$

$g_{mg2}r_{g2} \approx \mu_{g1-g2}$ なので, $g_{mg2}(r_{g2} // R_{g2}) \approx \mu_{g1-g2}R_{g2}/(r_{g2} + R_{g2})$ より,

$$\begin{aligned}
e_{g2} &\approx -\frac{\mu_{g1-g2}e_g}{2} \cdot \frac{R_{g2}}{r_{g2} + R_{g2}} \cdot \frac{1 + f \frac{I_p}{I_{g2}} \cdot \frac{R_L}{r_p + R_L}}{1 + f \frac{I_p}{I_{g2}} \cdot \frac{R_L}{r_p + R_L} \cdot \frac{R_{g2}}{r_{g2} + R_{g2}}} \\
&= -\frac{\mu_{g1-g2}e_g}{2} \cdot \frac{1 + f \frac{I_p}{I_{g2}} \cdot \frac{R_L}{r_p + R_L}}{\frac{r_{g2}}{R_{g2}} + 1 + f \frac{I_p}{I_{g2}} \cdot \frac{R_L}{r_p + R_L}}
\end{aligned} \tag{42}$$

両相の出力の和は,

$$-e_{p1} + e_{p2} = g_m e_g (r_p // R_L) \tag{43}$$

となり, 通常のカソード接地のゲインと一致します.

両相のバランスは,

$$\frac{e_{p2}}{-e_{p1}} = \frac{-\frac{e_{g2}}{\mu_{g1-g2}}}{e_g + \frac{e_{g2}}{\mu_{g1-g2}}} = \frac{-e_{g2}}{\mu_{g1-g2}e_g + e_{g2}} \tag{44}$$

より, $e_{g2} = -\mu_{g1-g2}e_g/2$ のときに完全になります. これは, 式(42)において, $R_{g2} \rightarrow \infty$ の場合に成り立ちます. すなわち, スクリーニンググリッドに定電流回路を入れてやればバランスが取れます.

4.1.1 数値例

$$\begin{aligned}
x &= f \frac{I_p 0}{I_{g20}} \cdot \frac{R_L}{r_p + R_L} = 0.6477 \times \frac{1.086}{0.231} \times \frac{142.3}{3009 + 142.3} = 0.137565 \\
A_x &= g_{mg2}(r_{g2} // R_{g2}) = 0.2735 \times (133.3 // 1000) = 32.162 \\
A_{g2} &= -\frac{A_x}{2} \cdot \frac{1 + x}{1 + \frac{x A_x}{\mu_{g1-g2}}} = -\frac{36.162}{2} \times \frac{1 + 0.137565}{1 + \frac{0.137565 \times 36.162}{36.57}} = -16.319 \\
A_{p1} &= -g_m \left(1 + \frac{A_{g2}}{\mu_{g1-g2}}\right) (r_p // R_L) = -1.286 \left(1 - \frac{16.319}{36.57}\right) (3009 // 142.3) = -96.78 \\
A_{p2} &= -g_m \frac{A_{g2}}{\mu_{g1-g2}} (r_p // R_L) = -1.286 \frac{-16.319}{36.57} (3009 // 142.3) = 77.99 \\
\frac{|A_{p2}|}{|A_{p1}|} &= \frac{77.99}{96.78} = 0.806 = -1.88 \text{ dB}
\end{aligned}$$

4.1.2 シミュレーション

6267_sgc.cir

```

1 6267 screen grid coupled phase invertor
2 .INCLUDE 6267.lib
3
4 VI 13 0 DC 0 AC 1

```

```
5 VI2 14 0 DC 0 AC 0
6 XV1 111 112 13 1 6267
7 XV2 121 122 14 1 6267
8 R1 13 0 1.5Meg
9 R2 91 12 1Meg
10 R3 91 22 1Meg
11 C1 12 22 0.1u
12 RK 1 0 762.5
13 CK 1 0 1000u
14 R5 11 91 180k
15 R6 21 91 180k
16 VIP1 11 111 DC 0
17 VIG21 12 112 DC 0
18 VIP2 21 121 DC 0
19 VIG22 22 122 DC 0
20 C2 11 31 0.1u
21 C3 21 41 0.1u
22 R7 31 0 680k
23 R9 41 0 680k
24
25 VB2 91 0 330
26
27 .control
28 ac dec 10 1k 1k
29 print abs(v(31)) abs(v(41)) abs(v(112))
30 .endc
31 .END
```

結果

```
1
2 Circuit: 6267 screen grid coupled phase inverter
3
4 abs(v(31)) = 9.677640e+01
5 abs(v(41)) = 7.798516e+01
6 abs(v(112)) = 1.631999e+01
```